Morse Code Translator

Jordan Wren
Patrick Buah

11/4/13

Hardware Used

Arduino Board

LED

270 ohm resistor

What is Morse Code?

Morse code is a method of transmitting text information as a series of on-off tones, lights, or clicks.

Used to be understood by any listener or observer without special equipment.

How does Morse Code Work?

Each character (letter or numeral) is represented by a unique sequence of dots and dashes.

The duration of a dash is three times the duration of a dot. Each dot or dash is followed by a short silence, equal to the dot duration.

The letters of a word are separated by a space equal to three dots (one dash), and the words are separated by a space equal to seven dots.

Table of Morse Code Letters



Implementation of Morse Code

We used two arrays of strings (one for letters and one for numerals) to keep track of our dots and dashes

So to find out what we need to flash for the first letter of the alphabet(A), we will get the string letters[0]

Example of Our Code

```
int ledPin = 12;

char* letters[] = {
  ".-", "-...", "-.-.", "-..", ".", "..-.", "--.", "....", "..",     // A-I
  ".---", "-.-", ".-..", "--", "-.", "---", ".--.", "--.-", ".-.",   // J-R
  "...", "-", "..-", "...-", ".--", "-..-", "-.--", "--.."           // S-Z
};

char* numbers[] = {"-----", ".----", "..---", "...--", "....-", ".....", "-....",
   "--...", "---..", "----."};

int dotDelay = 200;

void setup()
{
  pinMode(ledPin, OUTPUT);
  Serial.begin(9600);
}

void loop()
{
  char ch;
  if (Serial.available())       // is there anything to be read from USB?
  {
    ch = Serial.read();                 // read a single letter
    if (ch >= 'a' && ch <= 'z')
    {
      flashSequence(letters[ch - 'a']);
    }
    else if (ch >= 'A' && ch <= 'Z')
    {
      flashSequence(letters[ch - 'A']);
    }
    else if (ch >= '0' && ch <= '9')
    {
      flashSequence(numbers[ch - '0']);
    }
    else if (ch == ' ')
    {
     delay(dotDelay * 4);               // gap between words
    }
  }
}

void flashSequence(char* sequence)
{
    int i = 0;
    while (sequence[i] != NULL)
    {
        flashDotOrDash(sequence[i]);
        i++;
    }
    delay(dotDelay * 3);                // gap between letters
}

void flashDotOrDash(char dotOrDash)
{
  digitalWrite(ledPin, HIGH);
  if (dotOrDash == '.')
  {
    delay(dotDelay);
  }
  else // must be a -
  {
    delay(dotDelay * 3);
  }
  digitalWrite(ledPin, LOW);
  delay(dotDelay);                      // gap between flashes
}
```

Demonstration Time!